

Goblet Design

Douglas Meade, Ronda Sanders, and Xian Wu
Department of Mathematics

Instructions

You have been hired by a private company to design the most visually appealing goblet that meets the following criteria:

- the goblet will be molded using a symmetric mold, that is, the goblet must be a solid of revolution;
- the goblet must hold between 8 in³ and 10 in³ of liquid and use less than 6 in³ of glass;
- the height of the center of mass must be less than 3 times the radius of the foot so the goblet will be reasonably stable;
- thickness of the glass must be at least 0.08 in at its thinnest point.

Maple Essentials

Maple commands introduced for this project:

| Command | Description |
|------------------------|---|
| <code>piecewise</code> | define a piecewise-defined function The general syntax to represent $\begin{cases} f_1, & cond_1 \\ f_2, & cond_2 \\ \vdots & \vdots \\ f_n, & cond_n \end{cases}$ is: <code>piecewise(cond₁, f₁, cond₂, f₂, ..., cond_n, f_n);</code> where each $cond_i$ is an inequality and each f_i is an expression. |
| <code>solve</code> | solve an equation or system of equations <code>solve(eqn, var);</code> solves an equation, eqn , for one variable, var . <code>solve({eqn1, eqn2}, {var1, var2});</code> solves a system of two equations for two variables. |
| <code>assign</code> | <code>assign(values);</code> assigns the values returned by the <code>solve</code> command |

An Example in Maple:

Most maple commands used here were introduced in past labs. For those of you who have taken Maple in a Calculus I class here at USC, you may want to look back at the Roller Coaster project. Also, you may want to review the use of the `VolumeOfRevolution` command from last week's lab. For your convenience, this example is available in a Maple worksheet from our lab web page.

1. Basic Analyses: By studying our favorite goblets at home, we decide that our new goblet should have a foot (base) of 1.5 inches in radius and 0.5 inches in height. We would also like to have a slightly curved stem of 5 inches long and about 0.2 inches in radius. Finally, we want it to have a bowl that gradually extends to a radius of 1 inch in the first 1.5 inches from the stem, then to a radius of 0.9 inches in the next 0.5 inches, and to a radius of 1.6 inches in the last 1.5 inches. Put it in sideways and our goblet hence extends from $x = 0$ to $x = 9$. The outside profile of our goblet can therefore be described by a piecewise function that consists of three functions $f_1(x)$, $f_2(x)$, and $f_3(x)$ (corresponding to base, stem, and bowl sections of our goblet) defined over $0 \leq x < 0.5$, $0.5 \leq x < 5.5$, and $5.5 \leq x \leq 9$, respectively.
2. Working with Maple
 - (a) You should always put the command `restart` at the beginning of a Maple session as it clears the internal memory so that Maple acts (almost) as if just started. In general, you need to start over from the very beginning after changes are made. (You can use `!!!` to quickly re-execute the entire page.)
`> restart;`

- (b) Next, load needed packages using the `with` command.
- ```
> with(plots):
> with(Student[Calculus1]):
```
- (c) For the foot, we want to use a linear function that passes through the points  $(0, 1.5)$  and ends at  $(0.5, 0.2)$ . It can be computed by hand that  $f_1(x) = -2.6x + 1.5$  will do the job.
- ```
> f1:=x-> -2.6*x+1.5;
```
- (d) For a plain stem of radius 0.2, we may just set $f_2(x) = 0.2$. To have a slightly curved stem, we will add a slight wave to it using the sine function. For this particular goblet, we take the amplitude 0.05 as the height of the wave, adjust the period of the function to be π , and shift the function 0.5 to the right so we have $f_2(0.5) = f_1(0.5) = 0.2$. (Note: You can adjust the period of the sine function further to add more curves to the stem.)
- ```
> f2:=x-> 0.2+0.05*sin(2*(x-0.5));
```
- (e) By our design, our bowl needs to satisfy the following four conditions: It connects the stem at  $x = 5.5$  and takes values 1, 0.9, and 1.6 at  $x = 7$ ,  $x = 7.5$ , and  $x = 9$ , respectively. Therefore, let us define  $f_3(x) = ax^3 + bx^2 + cx + d$ , where  $a, b, c, d$  are four constants to be solved later to satisfy the above four requirements. Here are  $f_3(x)$  and 4 equations.
- ```
> f3:=x-> a*x^3+b*x^2+c*x+d;
> eq1:= f3(5.5)=f2(5.5);
> eq2:= f3(7)=1;
> eq3:= f3(7.5)=0.9;
> eq4:= f3(9)=1.6;
```
- (f) Now, let's solve for a, b, c , and d using `solve` command and assign solutions to *values*.
- ```
> values:=solve({eq1, eq2, eq3, eq4},{a, b, c, d});
```
- (g) We can then plug these solved values into our functions using the `assign` command.
- ```
> assign(values);
```
- (h) We are now ready to put all three functions together as a piecewise function and plot it to see the outside profile.
- ```
> F:=x->piecewise(x<0.5, f1(x), x>=0.5 and x<5.5, f2(x), x>=5.5, f3(x));
> plot(F(x), x=0..9, y=0..4, scaling=constrained);
```
- Note:** *The scaling=constrained insures that one unit on the x-axis equals one unit on the y-axis.*
- (i) To make sure that the thickness of the glass is at least 0.08 inches, we take the bowl part of the inside profile to be 0.1 inches less than the bowl part of the outside profile as follows:
- ```
> g3:= x->f3(x)-0.1;
> G:=x->piecewise(x<=5.5, 0, x>5.5, g3(x));
> plot(G(x), x=0..9, y=0..4, scaling=constrained);
```
- (j) Here is the goblet. Rotate and/or right click the plot to try some visual effect options!
- ```
> VolumeOfRevolution(F(x), G(x), x=0..9, scaling=constrained, output=plot,
orientation=[0,180], title='Example');
```
- (k) We still need to check the criteria. For the center of mass, we can see that it is located on the  $x$ -axis, say, at  $x = CM$ . It is not too hard to derive the following formula for  $CM$ :
- $$CM = \frac{\int_0^9 x(F^2(x) - G^2(x)) dx}{\int_0^9 (F^2(x) - G^2(x)) dx}.$$
- ```
> Capacity:= VolumeOfRevolution(G(x), 0, x=0..9);
> Glass:= VolumeOfRevolution(F(x), G(x), x=0..9);
> CM:=int(x*((F(x))^2-(G(x))^2), x=0..9)/int((F(x))^2-(G(x))^2, x=0..9);
> Ratio:=CM/F(0);
```

3. Conclusion: Looks like we have a stable goblet that satisfies the requirements. The visual appearance of the goblet can be improved. Now, it is up to you to design a more interesting and better looking goblet that satisfies a given set of criteria.